

## G11SBF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

G11SBF is a service routine which may be used prior to calling G11SAF to calculate the frequency distribution of a set of dichotomous score patterns.

### 2 Specification

```
SUBROUTINE G11SBF(IP, N, IS, X, NRX, IRL, IFAIL)
  INTEGER          IP, N, IS, NRX, IRL(N), IFAIL
  LOGICAL         X(NRX,IP)
```

### 3 Description

When each of  $n$  individuals responds to each of  $p$  dichotomous variables the data assumes the form of the matrix  $X$  defined below

$$X = \begin{bmatrix} x_{11} & x_{12} & \cdot & \cdot & \cdot & x_{1p} \\ x_{21} & x_{22} & \cdot & \cdot & \cdot & x_{2p} \\ \cdot & \cdot & & & & \cdot \\ \cdot & \cdot & & & & \cdot \\ x_{n1} & x_{n2} & \cdot & \cdot & \cdot & x_{np} \end{bmatrix} = \begin{bmatrix} \underline{x}'_1 \\ \underline{x}'_2 \\ \cdot \\ \cdot \\ \underline{x}'_n \end{bmatrix}$$

where the  $x$ 's take the value of 0 or 1 and  $\underline{x}_l = (x_{l1}, x_{l2}, \dots, x_{lp})'$ , for  $l = 1, 2, \dots, n$  denotes the score pattern of the  $l$ th individual (' denoting the transpose of a vector). G11SBF calculates the number of different score patterns,  $s$ , and the frequency with which each occurs. This information can then be passed to G11SAF.

### 4 References

None.

### 5 Parameters

- 1: IP — INTEGER *Input*  
*On entry:* the number of dichotomous variables,  $p$ .  
*Constraint:*  $IP \geq 3$ .
- 2: N — INTEGER *Input*  
*On entry:* the number of individuals in the sample,  $n$ .  
*Constraint:*  $N \geq 7$ .
- 3: IS — INTEGER *Output*  
*On exit:* the number of different score patterns,  $s$ .
- 4: X(NRX,IP) — LOGICAL array *Input/Output*  
*On entry:*  $X(i, j)$  must be set equal to `.TRUE.` if  $x_{ij} = 1$ , and `.FALSE.` if  $x_{ij} = 0$ , for  $i = 1, 2, \dots, n$ ;  $j = 1, 2, \dots, p$ .  
*On exit:* the first  $s$  rows of X contain the  $s$  different score patterns.

- 5:** NRX — INTEGER *Input*  
*On entry:* the first dimension of the array X as declared in the (sub)program from which G11SBF is called.  
*Constraint:*  $\text{NRX} \geq N$ .
- 6:** IRL(N) — INTEGER array *Output*  
*On exit:* the frequency with which the  $l$ th row of X occurs, for  $l = 1, 2, \dots, s$ .
- 7:** IFAIL — INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

On entry, IP < 3,  
 or N < 7,  
 or NRX < N.

## 7 Accuracy

Exact.

## 8 Further Comments

The time taken by the routine is small and increases with  $n$ .

## 9 Example

A program to count the frequencies of different score patterns in the following list:

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G11SBF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NRX, IPMAX
      PARAMETER        (NRX=100,IPMAX=5)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, IP, IS, J, N
*      .. Local Arrays ..
```

```

      INTEGER          IRL(NRX)
      LOGICAL          X(NRX,IPMAX)
*    .. External Subroutines ..
      EXTERNAL          G11SBF
*    .. Executable Statements ..
      WRITE (NOUT,*) 'G11SBF Example Program Results'
*    Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, IP
      IF (N.GT.0 .AND. N.LE.NRX .AND. IP.GT.0 .AND. IP.LE.IPMAX) THEN
        DO 20 I = 1, N
          READ (NIN,*) (X(I,J),J=1,IP)
20      CONTINUE
          IFAIL = 0
*
          CALL G11SBF(IP,N,IS,X,NRX,IRL,IFAIL)
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Frequency      Score pattern'
          WRITE (NOUT,*)
          DO 40 I = 1, IS
            WRITE (NOUT,99999) IRL(I), (X(I,J),J=1,IP)
40      CONTINUE
          END IF
          STOP
*
99999 FORMAT (1X,I5,12X,5L2)
      END

```

## 9.2 Program Data

G11SBF Example Program Data

```

10 3
F F F
F T F
T T T
F F F
F F T
F F F
F F F
T T F
F F T
F T T

```

## 9.3 Program Results

G11SBF Example Program Results

Frequency	Score pattern
4	F F F
1	F T F
1	T T T
2	F F T
1	T T F
1	F T T